
Embarcadero RAD Server (EMS) is a technology to create and deploy REST (micro) 

Services, and TMS WEB Core is a technology to create and deploy Single Page Web 

Applications (SPA). In this article, I’ll demonstrate how we can combine these two 

techniques for even more powerful applications. 

 

RAD Server (EMS) 

Let’s start by building a RAD Server (EMS) Micro Service using Delphi Enterprise. Do File | 

New – Other, and from the RAD Server (EMS) category double-click on the RAD Server 

(EMS) Package icon to create a new RAD Server (EMS) package. 

 

 
 

The wizard that follows gives us a choice of creating a new empty package, or a package with 

a resource. Although we can always add resources (even to an empty package), it’s easiest to 

create a package with a resource right from the start, so select the option to “Create package 

with resource”. 

 

http://www.drbob42.com/


 
 

If we click on the Next button, we can specify the name of the resource (which I’ve called 

WebCoreDemo for now). We can also determine the file type of the resource: either a normal 

unit, or a data module. The advantage of a data module is that we can use it as a container for 

non-visual components (for example for database access) without having to create and 

configure all components at run-time. 

 

So, for our demo, I’ve specified WebCoreEMS as name of the resource, and selected the Data 

Module as file type: 

 



 
 

The last page of the RAD Server (EMS) Package Wizard can be used to specify the endpoints 

that we want to expose. By default, the Get and GetItem endpoints are selected, 

corresponding to the GET HTTP verb. The Post endpoint corresponds to the POST HTTP 

verb, while PutItem and DeleteItem correspond to the PUT and DELETE HTTP verbs. 

 

For our demo, I do not only want to use the GET HTTP verb, but also demonstrate the use of 

the POST HTTP verb, so make sure to select Post as well as the default choices Get and 

GetItem. 

 



 
 

Clicking on Finish will generate the new EMS Package in an unnamed project with unnamed 

unit (by default Project1.bpl with Unit1.pas). I’ve renamed the project to WebCoreEmsDemo 

and the resource to WebCoreEmsResource. 

 



 
 

The WebCoreEmsResource unit contains the empty data module. We can plade data-access 

component on it, to get our hands on data from a database. Although RAD Server (EMS) is 

deployed with an InterBase database, we are not required to use that database for our other 

tables or queries. For this demo, I want to use a SQL Server database, specifically the SQL 

Server Northwind example database, and work with the Customers table. 

 

Delphi includes a number of data access technologies, the most recent and recommend one is 

FireDAC. However, when using SQL Server we can also use dbGo for ADO, a low-level way 

to work with the SQL Server database, without the need for additional deployment DLLs. 

 

SQL 

Place a TADOConnection component on the EMS data module, and configure the 

ConnectionString property to connect to the server machine with your database (in my case, 

that’s SQL Server Express with the Northwind example database). We also need a 

TADODataSet component called qrySelectCustomers to select all records from the Customers 

table, and a TADOCommand called qryUpdateCustomer component to update one customer 

record of the Customer table. 

 

 



 

The CommandType property of the qrySelectCustomers is set to cmdText, and the 

CommandText property contains the SQL query SELECT * FROM Customers 

 

The qryUpdateCustomer has the CommandType also set to cmdText and the CommandText 

assigned to the following SQL query: 
 

  UPDATE Customers 

     SET CompanyName = :CompanyName, 

         ContactName = :ContactName 

   WHERE CustomerID = :CustomerID 

 

This specific update statement will change the CompanyName and ContactName from the 

Customers table, for a give CustomerID. Of course, we can modify other fields as well, but 

this is just an example for the demo of this paper. The query has three parameters that need to 

be filled before we can execute it to update the Customer record. 

 

Customer 

Now it’s time to implement the Get, GetItem and Post methods of the WebCoreEMS 

resource. For the Get and GetItem, I want to produce pure JSON output that others can 

consume easily, including the TMS WEB Core client components. For that purpose, I’ve 

written a special unit that defines a class TCustomer with private fields (corresponding to the 

columns in the Customer table) and a method Read that takes a TDataSet and reads the 

columns from the dataset into the class instance fields. 
 

type 

  TCustomer = class 

  private 

    FCustomerID: String; 

    FCompanyName: String; 

    FContactName: String; 

    FContactTitle: String; 

    FAddress: String; 

    FCity: String; 

    FRegion: String; 

    FPostalCode: String; 

    FCountry: String; 

    FPhone: String; 

    FFax: String; 

  public 

    procedure Read(DS: TDataSet); 

  end; 

 

This is just for one customer, so we also need a TCustomers class to maintain an open array 

of customers, with a procedure to read the whole dataset into the array, plus a function to 

return the collection as one big JSON string: 
 

  TCustomers = class 

  public 

    Customers: array of TCustomer; 

  public 

    destructor Destroy; override; 

    procedure Read(DS: TDataSet); 

    function ToJSON: String; 

  end; 



 

The implementation is not very difficult. In order to read the fields from the dataset, we can 

use FieldByName for example, as follows: 
 

procedure TCustomer.Read(DS: TDataSet); 

begin 

  FCustomerID := DS.FieldByName('CustomerID').AsString; 

  FCompanyName := DS.FieldByName('CompanyName').AsString; 

  FContactName := DS.FieldByName('ContactName').AsString; 

  FContactTitle := DS.FieldByName('ContactTitle').AsString; 

  FAddress := DS.FieldByName('Address').AsString; 

  FCity := DS.FieldByName('City').AsString; 

  FRegion := DS.FieldByName('Region').AsString; 

  FPostalCode := DS.FieldByName('PostalCode').AsString; 

  FCountry := DS.FieldByName('Country').AsString; 

  FPhone := DS.FieldByName('Phone').AsString; 

  FFax := DS.FieldByName('Fax').AsString; 

end; 

 

Reading the entire TDataSet is just a loop around the Read call: 
 

procedure TCustomers.Read(DS: TDataSet); 

var 

  i: Integer; 

begin 

  SetLength(Customers, DS.RecordCount); 

  DS.First; 

  i := 0; 

  while not DS.Eof do 

  begin 

    Customers[i] := TCustomer.Create; 

    Customers[i].Read(DS); 

    Inc(i); 

    DS.Next; 

  end; 

end; 

 

Producing the JSON string is the result of the TJson.ObjectToJsonString from the 

REST.JSON unit: 
 

function TCustomers.ToJSON: String; 

begin 

  Result := TJson.ObjectToJsonString(Self) 

end; 

 

Get 

With these primitives available, we can implement the procedure Get as follows: 
 

procedure TWebCoreEMSResource1.Get(const AContext: TEndpointContext; 

  const ARequest: TEndpointRequest; const AResponse: TEndpointResponse); 

var 

  Custs: TCustomers; 

begin 

  qrySelectCustomers.Close; 

  qrySelectCustomers.Open; 

  Custs := TCustomers.Create; 

  try 

    Custs.Read(qrySelectCustomers); 



    AResponse.StatusCode := 200; 

    AResponse.Body.SetBytes(TEncoding.ASCII.GetBytes(Custs.ToJSON), 

      'application/json'); 

  finally 

    Custs.Free 

  end; 

end; 

 

The result of this method can be seen in a browser by calling the URL of the EMS Service, 

followed by the /WebCoreEMS resource specifier: http://localhost:8080/WebCoreEMS 

 

The (formatted) JSON output looks as follows: 
 

{ 

  "customers": [ 

    { 

      "customerID": "ALFKI", 

      "companyName": "Alfreds Flipperkast", 

      "contactName": "Maria abc 123123", 

      "contactTitle": "Sales Representative", 

      "address": "Obere Str. 57", 

      "city": "Berlin", 

      "region": "", 

      "postalCode": "12209", 

      "country": "Germany", 

      "phone": "030-0074321", 

      "fax": "030-0076545" 

    }, 

   ...  

} 

 

This is a JSON array that can be consumed by any self-respecting JSON client, including 

Delphi clients and TMS EMS Web Core clients. 

 

GetItem 

The GetItem method gets an Item parameter, to look at a specific Customer record. This item 

is usually the primary key or another unique identifier. In our case, it can be the CustomerID 

(which consists of a 5-character identifier for the customer record). 

The code to return the single customer can be as follows, using a simple filter to limit the 

dataset to just a single record: 
 

procedure TWebCoreEMSResource1.GetItem(const AContext: TEndpointContext; 

  const ARequest: TEndpointRequest; const AResponse: TEndpointResponse); 

var 

  Custs: TCustomers; 

  CustomerID: string; 

begin 

  CustomerID := ARequest.Params.Values['item']; 

  qrySelectCustomers.Close; 

  qrySelectCustomers.Open; 

  qrySelectCustomers.Filter := 'CustomerID = ''' +  CustomerID + ''''; 

  qrySelectCustomers.Filtered := True; 

  Custs := TCustomers.Create; 

  try 

    Custs.Read(qrySelectCustomers); 

    AResponse.StatusCode := 200; 

    AResponse.Body.SetBytes(TEncoding.ASCII.GetBytes(Custs.ToJSON), 



      'application/json'); 

  finally 

    qrySelectCustomers.Filtered := False; 

    Custs.Free 

  end; 

end; 

 

The result of the GetItem method can be see by adding a CustomerID to the previous URL, 

for example as follows: http://localhost:8080/WebCoreEMS/ANTON 

 

 
 

Similar output compared to the Get, but this time only the result of a single customer. 

 

Post 

The update of a single Customer record is done using the query with 3 parameters. This 

means we have to modify the method definition to bind the names of the parameters to the 

vaules. The definition of the GetItem method was as follows: 
 

    [ResourceSuffix('{item}')] 

    procedure GetItem(const AContext: TEndpointContext; 

      const ARequest: TEndpointRequest; 

      const AResponse: TEndpointResponse); 

 

Based on this, we can define the Post method as follows: 
 

    

[ResourceSuffix('UpdateCustomer/{CustomerID}/{CompanyName}/{ContactName}')] 

    procedure GetUpdateCustomer(const AContext: TEndpointContext; 

      const ARequest: TEndpointRequest; 

      const AResponse: TEndpointResponse); 

 

The implementation is straightforward: 



 

procedure TWebCoreEMSResource1.Post(const AContext: TEndpointContext; 

  const ARequest: TEndpointRequest; const AResponse: TEndpointResponse); 

begin 

  qryUpdateCustomer.Parameters.ParamByName('CustomerID').Value := 

    ARequest.Params.Values['CustomerID']; 

  qryUpdateCustomer.Parameters.ParamByName('CompanyName').Value := 

    ARequest.Params.Values['CompanyName']; 

  qryUpdateCustomer.Parameters.ParamByName('ContactName').Value := 

    ARequest.Params.Values['ContactName']; 

  try 

    qryUpdateCustomer.Execute; 

    AResponse.StatusCode := 200; 

    AResponse.Body.SetBytes(BytesOf('OK'), 'text/plain'); 

  except 

    on E: Exception do 

      AResponse.Body.SetBytes(BytesOf(E.Message), 'text/plain'); 

  end; 

end; 

 

We can test the WebCoreEmsDemo using the EMSDevServer.exe for development, but for 

real deployment, we need to use the EMSServer.dll ISAPI DLL and deploy the 

WebCoreEmsDemo.bpl on a web server (using IIS on Windows for example). 

 

Deployment 

Although you can use the EMSDevServer.exe as development and test environment, you 

really need to deploy the WebCoreEmsDemo.bpl using the ISAPI DLL version of the 

EMSServer.dll. In fact, I had to deploy a number of files in order to make the 

WebCoreEmsDemo.dpl work: 
 

adortl250.bpl 

bindcomp250.bpl 

bindengine250.bpl 

borlndmm.dll 

CustomIPTransport250.bpl 

dbrtl250.bpl 

emsclient250.bpl 

EMSConsole.dll 

emsedge250.bpl 

emshosting250.bpl 

EMSServer.dll 

emsserver.ini 

emsserverapi250.bpl 

FireDAC250.bpl 

FireDACCommon250.bpl 

FireDACCommonDriver250.bpl 

FireDACIBDriver250.bpl 

FireDACSqliteDriver250.bpl 

inet250.bpl 

RESTComponents250.bpl 

rtl250.bpl 

soaprtl250.bpl 

vcl250.bpl 

vcldb250.bpl 

vclFireDAC250.bpl 

vclx250.bpl 

WebCoreEmsDemo.dpl 

xmlrtl250.bpl 



 

We also need to modify the emsserver.ini to include the location of our EMS web service 

package, in the Server.Packages section: 
 

[Server.Packages] 

;# This section is for extension packages. 

;# Extension packages are used to register custom resource endpoints 

;c:\mypackages\basicextensions.bpl=mypackage description 

c:\inetpub\wwwroot\Deployment\EMS\WebCoreEmsDemo.bpl=WebCoreEMS 

 

TMS WEB Core 

Once the WebCoreEmsDemo is deployed, we can write the TMS WEB Core client to connect 

to it and work with the data from the EMS micro service. Feel free to use the deployed 

version of the WebCoreEmsDemo.bpl as deployed on my web server with my deployment 

license of RAD Server (EMS). 

 

Before we add the TMS WEB Core project, we may first want to configure the TMS WEB 

options. Go to Tools | Options and select the TMS WEB options. 

 

 
 

Note the Output Path, by default set to .\$(Platform)\$(Config), and the URL, by default set to 

http://localhost:8000/$(ProjectName). These values are OK for the TMS Web Server, but not 

for real-world deployment on IIS. In that case, I want to change the Output Path to 

c:\inetpub\wwwroot\$(ProjectName) and the URL to http://localhost/$(ProjectName) without 

the 8000 value for the port. 

 

http://localhost:8000/$(ProjectName)
http://localhost/$(ProjectName)


TMS WEB Core Application 

To add a TMS WEB Core project to the project group, do File | New – Other again, this time 

moving to the TMS Web category. Two wizards are available here: TMS Web Application 

and TMS Web Console Application. 

 

 
 

This will produce a new project with a html project file as well as a .pas file with associating 

.dfm and .html files. I’ve saved the project in WebCoreEmsClient.dpr, the project HTML file 

in index.html, and the page in SpaForm.pas (with corresponding .dfm and .html files). 

 



 
 

This is just an “empty” project for now, but it will generate the JavaScript file as soon as we 

compile the project. Pressing F9 will produce the following files in the specified location in 

c:\inetpub\wwwroot\webCoreEmsClient 

 

index.html 

SpaForm.html 

WebCoreEmsClient.js 

WebCoreEmsClient.js.map 

 

The file index.html is the starting point, and will show up if we open a browser at 

http://localhost/webCoreEmsClient (right now showing an empty page, but we will fill it up 

with data from the EMS micro service shortly). 

 

Web Client Data 

In order to connect to a REST server, and get our hands on JSON data to be able to process 

and display inside the WEB Core Form, we need to place three non-visual components from 

the TMS Web DB category of the Tool Palette. 

First, a WebClientConnection component, with the URI pointing to your RAD Server (EMS) 

URI, or if you want to connect to my ready-to-user server, point it to 

http://localhost/webCoreEmsClient


https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS and set the DataNode property to 

customers to get the records from that array node. 

Second, we need a TWebClientDataSet component, with the Connection property pointing to 

the WebClientConnection1 component. 

Third, we need a TWebClientDataSource component, with the DataSet property pointing to 

the TWebClientDataSet1 component. 

 

This is all very similar to regular VCL data-access and data-aware components, so should not 

be that difficult to understand. 

The first visual component we can now place, is a TWebDBNavigator component, with the 

DataSource property obviously pointing to the TWebClientDataSource1 component. 

 

 
 

Apart from the TWebDBNavigator, there are a dozen more TMS Web DB components that 

we can use to build our web user interface, including the useful TWebDBEdit. 

 

https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS


 
 

For my demo application, I have placed one TWebDBLabel and two TWebDBEdit 

components on the form, called them resp. lbCustomerID , edCompanyName, 

edContactName, and connected their DataSource property to the WebClientDataSource1 

component. 

It is tempting to open up the DataField property as well, to connect these data-aware TWeb 

components to the actual data, but that’s not possible at design-time. Not even if we have set 

the Active property of the WebClientConnection1 component to True. I’m afraid we have to 

set the DataField properties at runtime. 
 

    lbCustomerID.DataField := 'CustomerID'; 

    edCompanyName.DataField := 'CompanyName'; 

    edContactName.DataField := 'ContactName'; 

 

At the same time, we should add persistent fields to the WebClientDataSet, and in fact in 

order to demonstrate how we set the URI of the WebClientConnection, and do everything that 

needs to be done, I’ve written the following code in the OnClick event handler of a 

TWebButton component: 
 

procedure TForm1.WebButton1Click(Sender: TObject); 

begin 

  WebMemo1.Lines.Clear; 

  try 

    WebClientConnection1.Active := False; 



    WebClientDataSet1.Close; 

    WebClientConnection1.URI := 

       'https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS'; 

    WebClientConnection1.DataNode := 'customers'; 

    WebClientDataSet1.FieldDefs.Clear; 

    WebClientDataSet1.FieldDefs.Add('customerID',ftstring,5); // case sensitive 

    WebClientDataSet1.FieldDefs.Add('companyName',ftstring,40); 

    WebClientDataSet1.FieldDefs.Add('contactName',ftstring,30); 

    lbCustomerID.DataField := 'CustomerID'; 

    edCompanyName.DataField := 'CompanyName'; 

    edContactName.DataField := 'ContactName'; 

    WebClientConnection1.Active := True; 

  except 

    on E: Exception do 

      WebMemo1.Lines.Add(E.Message); 

  end; 

end; 

 

A TWebMemo component is added to display some trace and error messages that we may 

encounter along the way: 
 

 
 

Running this application, and clicking on the Open button, will indeed show the data in the 

label and two edit controls: 

 

https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS


 
 

Note that this data is coming from the RAD Server (EMS) micro service deployed on my 

server at this time. And you may notice some edits in the data, made by other people who also 

connected to the WebCoreEmsDemo package. We will also be able to make modifications to 

the data, but let’s first examine the events that can happen in the TMS WEB Core data access 

components. 
 

Event Handlers 

The TWebClientConnection component has three event handlers that we can hook into: 

BeforeConnect, AfterConnect, and OnConnectError. I find it instructive to log these events in 

the memo control, as follows: 
 

procedure TFormEmsDemo.WebClientConnection1BeforeConnect(Sender: TObject); 

begin 

  WebMemo1.Lines.Add('BeforeConnect'); 

end; 

 

procedure TFormEmsDemo.WebClientConnection1AfterConnect(Sender: TObject); 

begin 

  WebMemo1.Lines.Add('AfterConnect'); 

end; 

 

procedure TFormEmsDemo.WebClientConnection1ConnectError(Sender: TObject; 

  ErrorCode: Integer); 

begin 

  WebMemo1.Lines.Add('ConnectError ' + IntToStr(ErrorCode)); 

end; 



 

The TWebClientDataSet has no event handlers, but there are three event handlers in the 

TWebClientDataSource that we can use: OnDataChange, OnStateChange and OnUpdateData. 

I have logged these in the same way: 
 

procedure TFormEmsDemo.WebClientDataSource1DataChange(Sender: TObject; 

  Field: TField); 

begin 

  WebMemo1.Lines.Add('DataChange'); 

end; 

 

procedure TFormEmsDemo.WebClientDataSource1StateChange(Sender: TObject); 

begin 

  WebMemo1.Lines.Add('StateChange'); 

end; 

 

procedure TFormEmsDemo.WebClientDataSource1UpdateData(Sender: TObject); 

begin 

   WebMemo1.Lines.Add('UpdateData'); 

end; 

 

It is interesting to see when these events are fires when we navigate through the data using the 

TWebDBNavigator. If we open the dataset (click on Open), we get the BeforeConnect and 

AfterConnect, plus the  StateChange and DataChange of the WebClientDataSource. 
 

BeforeConnect 

AfterConnect 

StateChange 

DataChange 

 

Now, if I enter the edContactName control and start to type, we get another StateChange 

event (dataset goes in Edit mode) as well as a DataChange event because I modified the data 

in the TWebDBEdit control. 
 

StateChange 

DataChange 

 

If I now click on the Post button of the TWebDBNavigator, I get an UpdateData, a 

StateChange and a DataChange event. 
 

UpdateData 

StateChange 

DataChange 

 

That’s ideal, because this means we can use the UpdateData event handler to send the update 

– which is in our local TWebClientDataSet only – back to the RAD Server (EMS) micro 

service, so the update ends up in the actual database. 

 

However.. if I do not click on the Post button of the TWebDBNavigator, but just click on the 

Next button to move to the next record, we get a different set of events. In that case, we get 

two UpdateData events, followed by StateChange (back to browse mode), DataChange and 

another DataChange: 
 

UpdateData 

UpdateData 

StateChange 

DataChange 



DataChange 

 

The second call to UpdateData may be too much, and we can take care of that by adding a 

protected Boolean field called TableUpdating to the web form, set to True when we enter the 

UpdateData method, and ensureing we do not re-enter the same method. 
 

  protected 

    TableUpdating: Boolean; 

 

We need to set if to False when the form is created: 
 

procedure TForm1.WebFormCreate(Sender: TObject); 

begin 

  TableUpdating := False; 

end; 

 

The implementation of UpdateData now becomes as follows: 
 

procedure TFormEmsDemo.WebClientDataSource1UpdateData(Sender: TObject); 

begin 

  if not TableUpdating then 

  try 

    TableUpdating := True; 

    WebMemo1.Lines.Add('UpdateData'); 

    if WebClientDataSet1.State in dsEditModes then 

    begin 

      WebClientDataSet1.Post; 

      WebMemo1.Lines.Add('Post'); 

    end; 

 

  finally 

    TableUpdating := False; 

  end; 

end; 

 

This will ensure that the update is written to the dataset, although not to the REST Server, yet. 

 

TMS Web REST 

In order to send the update of the customer to the RAD Server (EMS) micro service, we must 

call the WebCoreEmsDemo microservice with the POST protocol, passing the CustomerID, 

CompanyName and ContactName as arguments. An example URL would be as follows: 
 

URL := 'https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS/UpdateCustomer' + 

  '/' + lbCustomerID.Caption + 

  '/' + edCompanyName.Text + 

  '/' + edContactName.Text; 

 

Note that we cannot test this URL in the browser, because we need to POST it, and inside the 

browser we can only send GET requests. 

 

In order to send Web REST requests from within a TMS WEB Core application, we can use 

the TWebHTTPRequest component, so place one on the Web Form. 

 



 
 

First, we need to ensure that we POST the HTTP REST request, so make sure to set the 

Command property to httpPOST. 

Then, we can set the URL property to the one we defined on the previous page, and call the 

Execute method to send the HTTP REST request to the RAD Server (EMS) micro service. If 

we do this right after the Post of the WebClientDataSet1, then the entire UpdateData event 

handler becomes as follows: 

 
procedure TForm1.WebClientDataSource1UpdateData(Sender: TObject); 

begin 

  if not TableUpdating then 

  try 

    TableUpdating := True; 

    WebMemo1.Lines.Add('UpdateData'); 

    if WebClientDataSet1.State in dsEditModes then 

    begin 

      WebClientDataSet1.Post; 

      WebMemo1.Lines.Add('Post'); 

    end; 

    WebHttpRequest1.URL := 

      'https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS/UpdateCustomer' + 

      '/' + lbCustomerID.Caption + 

      '/' + edCompanyName.Text + 

      '/' + edContactName.Text; 

    WebMemo1.Lines.Add(WebHttpRequest1.URL); 

    WebHttpRequest1.Execute; 

  finally 

    TableUpdating := False; 

  end; 

end; 

https://www.bobswart.nl/ems/emsserver.dll/WebCoreEMS/UpdateCustomer


 

The example source code contains both the links to the localhost:8000 version of the RAD 

Server (EMS) micro service, and the deployed version on my web server. Feel free to make 

your own modifications, and let me know if you have any questions or comments. 

 

For more information or support regarding Delphi, RAD Server (EMS) or TMS WEB Core, 

feel free to contact the author: Bob Swart from Bob Swart Training & Consultancy (eBob42) 

at Bob@eBob42.com or via https://www.linkedin.com/in/drbob42  

